This Maple worksheet is denoted sirs3c.
It is used to study an SIRS model with demography, for R0>1 and α1 large.
The deterministic version of the model then shows damped oscillations toward an endemic infection level.
The particular model dealt with here has a state space with 3 variables: S, I, and R, and the infection rate is "conventional", with N in the denominator.
Two things are done here.
First we derive an expression for the angular frequency of the deterministic model oscillations, and after that we give a derivation of the moments of a diffusion approximation.
Ingemar Nåsell, KTH, Stockholm, 2012-10-03.

```
> restart;
  with(LinearAlgebra,Transpose,Eigenvalues,
  CharacteristicPolynomial);
  with(VectorCalculus,Jacobian);
  interface(imaginaryunit=II);
```

$$[Transpose, Eigenvalues, CharacteristicPolynomial]$$

$$[Jacobian]$$

$$I \quad (1)$$

The reason for changing the notation used for the imaginary unit is that "I" will be used below to denote the number of infected individuals.
The original transition rates are stored in the table transA:

```
> transA:=table([[1,0,0]=mu*N,[1,0,-1]=delta*R,[-1,1,0]=beta*S*
  I/N,[-1,0,0]=mu*S,[0,-1,1]=gamma*I,[0,-1,0]=mu*I,[0,0,-1]=mu*R]
  ):
```

After scaling: x1=S/N, x2=I/N, x3=R/N, and reparametrization R0=β/(γ+μ), α1= (γ+μ)/μ, α2=(δ+μ)/μ, we get: S=x1*N, I=x2*N, R=x3*N, β=μ*α1*R0, γ=μ*(α1−1), δ=μ*(α2−1).
The Maple procedure "scale" is used to change the transition rates accordingly:

```
> scale:=proc(tab)
    local xA,n,xB,xC;
    xA:=op(2,eval(tab));
    n:=nops(xA);
    xB:=subs(S=x1*N,I=x2*N,R=x3*N,beta=mu*alpha1*R0,gamma=mu*
  (alpha1-1),delta=mu*(alpha2-1),xA);
    xC:=[seq(lhs(op(i,xB))=simplify(rhs(op(i,xB)/N)),i=1..n)];
    table(xC);
  end proc:
```

Apply the scaling and reparametrization described above to get the table of transition rates "trans":

```
> trans:=scale(transA);
```

$$trans := table([[0, 0, -1] = \mu\, x3, [0, -1, 0] = \mu\, x2, [0, -1, 1] = \mu\,(\alpha1 - 1)\, x2, [-1, 0, 0] = \mu\, x1, [-1, 1, 0] = \mu\,\alpha1\, R0\, x1\, x2, [1, 0, -1] = \mu\,(\alpha2 - 1)\, x3, [1, 0, 0] = \mu]) \quad (2)$$

Next is a procedure that determines the right-hand sides of the deterministic ODEs for the scaled variables x1, x2, x3 from the table of transition rates:

```
> equ:=proc(i,tab)
    local x,n;
    x:=op(2,eval(tab));
    add(lhs(x[n])[i]*rhs(x[n]),n=1..nops(x));
  end proc:
```

The 3 right-hand sides are as follows:

```
> eq1:=equ(1,trans);
  eq2:=simplify(equ(2,trans));
  eq3:=simplify(equ(3,trans));
```

$$eq1 := -\mu\, x1 - \mu\, \alpha 1\, R0\, x1\, x2 + \mu\, (\alpha 2 - 1)\, x3 + \mu$$

$$eq2 := -\mu\, x2\, \alpha 1 + \mu\, \alpha 1\, R0\, x1\, x2$$

$$eq3 := \mu\, x2\, \alpha 1 - \mu\, x2 - \mu\, x3\, \alpha 2 \qquad \textbf{(3)}$$

The critical points:

```
> crit:=solve({eq1,eq2,eq3},{x1,x2,x3});
```

$$crit := \{x1 = 1,\ x2 = 0,\ x3 = 0\},\ \left\{x1 = \frac{1}{R0},\ x2 = \frac{\alpha 2\,(-1 + R0)}{R0\,(\alpha 2 + \alpha 1 - 1)},\ x3 = \frac{(-1 + R0)\,(\alpha 1 - 1)}{R0\,(\alpha 2 + \alpha 1 - 1)}\right\} \qquad \textbf{(4)}$$

The point corresponding to an endemic infection level is termed (x10,x20,x30):

```
> x10:=rhs(crit[2][1]);
  x20:=rhs(crit[2][2]);
  x30:=rhs(crit[2][3]);
```

$$x10 := \frac{1}{R0}$$

$$x20 := \frac{\alpha 2\,(-1 + R0)}{R0\,(\alpha 2 + \alpha 1 - 1)}$$

$$x30 := \frac{(-1 + R0)\,(\alpha 1 - 1)}{R0\,(\alpha 2 + \alpha 1 - 1)} \qquad \textbf{(5)}$$

The Jacobian of the system of ODEs is denoted Bx:

```
> Bx:=Jacobian([eq1,eq2,eq3],[x1,x2,x3]);
```

$$Bx := \begin{bmatrix} -\mu - \mu\, x2\, \alpha 1\, R0 & -\mu\, \alpha 1\, R0\, x1 & \mu\,(\alpha 2 - 1) \\ \mu\, x2\, \alpha 1\, R0 & -\alpha 1\, \mu + \mu\, \alpha 1\, R0\, x1 & 0 \\ 0 & \alpha 1\, \mu - \mu & -\mu\, \alpha 2 \end{bmatrix} \qquad \textbf{(6)}$$

Evaluate the Jacobian at the critical point:

```
> B:=simplify(subs(x1=x10,x2=x20,x3=x30,Bx));
```

(7)

$$B := \begin{bmatrix} -\frac{\mu\,(\alpha 2+\alpha 1-1-\alpha 1\,\alpha 2+\alpha 1\,\alpha 2\,R0)}{\alpha 2+\alpha 1-1} & -\alpha 1\,\mu & \mu\,(\alpha 2-1) \\ \frac{\mu\,\alpha 2\,(-1+R0)\,\alpha 1}{\alpha 2+\alpha 1-1} & 0 & 0 \\ 0 & \alpha 1\,\mu-\mu & -\mu\,\alpha 2 \end{bmatrix} \quad (7)$$

We proceed to determine the eigenvalues of the matrix B.
We use first the command "Eigenvalues" to show that one of the eigenvalues equals -μ.
After that, we use the command "CharacteristicPolynomial" and the knowledge that one eigenvalue equals −μ to derive a quadratic equation for the two remaining eigenvalues.

```
> eig:=Eigenvalues(B);
```

$$eig := \Big[[-\mu], \Big[\frac{1}{2}\,\frac{1}{\alpha 2+\alpha 1-1}\Big(\Big(-\alpha 2^2-\alpha 1\,\alpha 2\,R0+\alpha 2 + \big(\alpha 2^4-2\,\alpha 1\,\alpha 2^3\,R0-2\,\alpha 2^3+\alpha 1^2\,\alpha 2^2\,R0^2+6\,\alpha 1\,\alpha 2^2\,R0+\alpha 2^2 - 8\,\alpha 2^2\,\alpha 1+4\,\alpha 2^3\,\alpha 1+8\,\alpha 2^2\,\alpha 1^2-8\,\alpha 1^2\,\alpha 2^2\,R0-8\,\alpha 2\,\alpha 1^2+4\,\alpha 2\,\alpha 1^3 - 4\,\alpha 2\,\alpha 1^3\,R0+8\,\alpha 1^2\,\alpha 2\,R0+4\,\alpha 1\,\alpha 2-4\,\alpha 1\,\alpha 2\,R0\big)^{1/2}\Big)\,\mu\Big)\Big],$$
$$\Big[-\frac{1}{2}\,\frac{1}{\alpha 2+\alpha 1-1}\Big(\Big(\alpha 2^2+\alpha 1\,\alpha 2\,R0-\alpha 2 + \big(\alpha 2^4-2\,\alpha 1\,\alpha 2^3\,R0-2\,\alpha 2^3+\alpha 1^2\,\alpha 2^2\,R0^2+6\,\alpha 1\,\alpha 2^2\,R0+\alpha 2^2 - 8\,\alpha 2^2\,\alpha 1+4\,\alpha 2^3\,\alpha 1+8\,\alpha 2^2\,\alpha 1^2-8\,\alpha 1^2\,\alpha 2^2\,R0-8\,\alpha 2\,\alpha 1^2+4\,\alpha 2\,\alpha 1^3 - 4\,\alpha 2\,\alpha 1^3\,R0+8\,\alpha 1^2\,\alpha 2\,R0+4\,\alpha 1\,\alpha 2-4\,\alpha 1\,\alpha 2\,R0\big)^{1/2}\Big)\,\mu\Big)\Big]\Big] \quad (8)$$

One of the eigenvalues is thus seen to be equal to −μ.
Next we determine the characteristic polynomial p:

```
> p:=CharacteristicPolynomial(B,lambda);
```

$$p := \lambda^3+\frac{\mu\,(\alpha 2^2+\alpha 1-1+\alpha 1\,\alpha 2\,R0)\,\lambda^2}{\alpha 2+\alpha 1-1} + \frac{\alpha 2\,\mu^2\,(\alpha 2+\alpha 1-1-\alpha 1\,\alpha 2+\alpha 1\,\alpha 2\,R0-\alpha 1^2+\alpha 1^2\,R0)\,\lambda}{\alpha 2+\alpha 1-1}+\mu^3\,\alpha 2\,(-1+R0)\,\alpha 1 \quad (9)$$

To proceed, we derive a quadratic equation for the remaining two eigenvalues by dividing p by λ+μ, and simplifying:

```
> p1:=map(simplify,collect(simplify(p/(lambda+mu)),lambda));
```

$$p1 := \lambda^2 + \frac{\mu\,\alpha 2\,(\alpha 2+\alpha 1\,R0-1)\,\lambda}{\alpha 2+\alpha 1-1} + \alpha 1\,(-1+R0)\,\alpha 2\,\mu^2 \quad (10)$$

The two roots of p1=0, where p1 = $\lambda^2 + a\cdot\lambda + b$, can be written $-a/2 \pm i * \Omega$, with $\Omega = \sqrt{b-\left(\frac{a}{2}\right)^2} = \mu\cdot\sqrt{\alpha 1\cdot\alpha 2\cdot(R0-1)-\left(\frac{\alpha 2\cdot R1}{2}\right)^2}$, where R1 = $\frac{\alpha 1\cdot R0+\alpha 2-1}{\alpha 1+\alpha 2-1}$.
This concludes the derivation of an expression for the angular frequency.
Next we turn toward a derivation of a diffusion approximation.
Covariances of x[i]x[j] are determined by cov1:

```
> cov1:=proc(i,j,tab)
    local x,n;
    x:=op(2,eval(tab));
    add(lhs(x[n])[i]*lhs(x[n])[j]*rhs(x[n]),n=1..nops(x));
  end proc:
```

The local covariance matrix S is determined by the procedure cov:

```
> cov:=proc(tab)
    local i,j,d,S;
    d:=nops(lhs(op(2,eval(tab))[1]));
    for i from 1 to d do
      for j from 1 to d do
        S[i,j]:=cov1(i,j,tab);
      od;
    od;
    S:=Matrix(d,S);
  end proc:
```

By using the table of transition rates in "trans", we get

```
> Sx:=simplify(cov(trans));
```

$$Sx := [[\mu\,x1+\mu\,\alpha 1\,R0\,x1\,x2+\mu\,x3\,\alpha 2-\mu\,x3+\mu,\ -\mu\,\alpha 1\,R0\,x1\,x2,\ -\mu\,(\alpha 2-1)\,x3],\ [-\mu\,\alpha 1\,R0\,x1\,x2,\ \mu\,x2\,\alpha 1+\mu\,\alpha 1\,R0\,x1\,x2,\ -\mu\,(\alpha 1-1)\,x2],\ [-\mu\,(\alpha 2-1)\,x3,\ -\mu\,(\alpha 1-1)\,x2,\ \mu\,x2\,\alpha 1-\mu\,x2+\mu\,x3\,\alpha 2]] \quad (11)$$

Evaluate the local covariance matrix at the critical point:

```
> S:=simplify(subs(x1=x10,x2=x20,x3=x30,Sx));
```

$$S := \left[\left[\frac{2\,\mu\,(\alpha 2+\alpha 1-1-\alpha 1\,\alpha 2+\alpha 1\,\alpha 2\,R0)}{R0\,(\alpha 2+\alpha 1-1)},\ -\frac{\mu\,\alpha 2\,(-1+R0)\,\alpha 1}{R0\,(\alpha 2+\alpha 1-1)},\ -\frac{\mu\,(\alpha 2-1)\,(-1+R0)\,(\alpha 1-1)}{R0\,(\alpha 2+\alpha 1-1)}\right],\ \left[-\frac{\mu\,\alpha 2\,(-1+R0)\,\alpha 1}{R0\,(\alpha 2+\alpha 1-1)},\ \frac{2\,\mu\,\alpha 2\,(-1+R0)\,\alpha 1}{R0\,(\alpha 2+\alpha 1-1)},\ -\frac{\mu\,(\alpha 1-1)\,\alpha 2\,(-1+R0)}{R0\,(\alpha 2+\alpha 1-1)}\right],\ \left[-\frac{\mu\,(\alpha 2-1)\,(-1+R0)\,(\alpha 1-1)}{R0\,(\alpha 2+\alpha 1-1)},\ -\frac{\mu\,(\alpha 1-1)\,\alpha 2\,(-1+R0)}{R0\,(\alpha 2+\alpha 1-1)},\right.\quad (12)$$

$$\frac{2\,\mu\,(\alpha 1-1)\,\alpha 2\,(-1+R0)}{R0\,(\alpha 2+\alpha 1-1)}\Bigg]\Bigg]$$

Now proceed to solve A=-S, where A=B*SIG+SIG*BT, and where BT=Transpose(B).
First introduce notation for the elements of the matrix SIG:

```
> SIG:=Matrix(3,[s11,s12,s13,s21,s22,s23,s31,s32,s33]);
```

$$SIG := \begin{bmatrix} s11 & s12 & s13 \\ s21 & s22 & s23 \\ s31 & s32 & s33 \end{bmatrix} \tag{13}$$

Next determine the matrix A:

```
> A:=Matrix(evalm(B&*SIG+SIG&*Transpose(B))):
```

Solve the 9 scalar equations found from the matrix equation A+S=0 for the 9 unknowns in the matrix SIG:

```
> solve(convert(A+S,set),convert(SIG,set)):
```

```
> assign(%);
```

Determine the first term of the asymptotic approximation of each of the elements of the matrix SIG as α1 becomes large.

```
> s11a:=simplify(op(1,asympt(s11,alpha1)));
  s12:=s12;
  s13a:=simplify(op(1,asympt(s13,alpha1)));
  s21:=s21;
  s22a:=simplify(op(1,asympt(s22,alpha1)));
  s23a:=simplify(op(1,asympt(s23,alpha1)));
  s31a:=simplify(op(1,asympt(s31,alpha1)));
  s32a:=simplify(op(1,asympt(s32,alpha1)));
  s33a:=simplify(op(1,asympt(s33,alpha1)));
```

$$s11a := \frac{\alpha 1}{\alpha 2\, R0^2}$$

$$s12 := -\frac{1}{R0}$$

$$s13a := -\frac{\alpha 1}{\alpha 2\, R0^2}$$

$$s21 := -\frac{1}{R0}$$

$$s22a := \frac{-1+R0}{R0^2}$$

$$s23a := \frac{1}{R0^2}$$

$$s31a := -\frac{\alpha 1}{\alpha 2\, R0^2}$$

$$s32a := \frac{1}{R0^2}$$

$$s33a := \frac{\alpha 1}{\alpha 2\, R0^2} \qquad (14)$$

The expectation of S+I+R, divided by N, is denoted Expsum:

```
> Expsum:=simplify(x10+x20+x30);
```

$$Expsum := 1 \qquad (15)$$

The variance of S+I+R, divided by N, is denoted Varsum:

```
> Varsum:=simplify(s11+s22+s33+2*s12+2*s13+2*s23);
```

$$Varsum := 1 \qquad (16)$$

mu_I = x20*N divided by sigma_I=sqrt(N*s22) is called ρI. Determine an approximation of ρI for large α1:

```
> x20;
  x20a:=op(1,asympt(x20,alpha1));
  rhoI:=simplify(x20a*N/sqrt(s22a*N)) assuming R0>1;
```

$$\frac{\alpha 2\,(-1+R0)}{R0\,(\alpha 2+\alpha 1-1)}$$

$$x20a := \frac{\alpha 2\,(-1+R0)}{R0\,\alpha 1}$$

$$rhoI := \frac{\alpha 2\sqrt{-1+R0}\,\sqrt{N}}{\alpha 1} \qquad (17)$$

```
> x10;
  rhoS:=simplify(x10*N/sqrt(s11a*N)) assuming R0>1,alpha1>1,
  alpha2>1;
```

$$\frac{1}{R0}$$

$$rhoS := \frac{\sqrt{N}\,\sqrt{\alpha 2}}{\sqrt{\alpha 1}} \qquad (18)$$

```
> x30;
```

$$\frac{(-1+R0)\,(\alpha 1-1)}{R0\,(\alpha 2+\alpha 1-1)} \qquad (19)$$

```
> x30a:=op(1,asympt(x30,alpha1));
```

$$x30a := \frac{-1+R0}{R0} \qquad (20)$$

```
> rhoR:=simplify(x30a*N/sqrt(s33a*N)) assuming R0>1,N>0,alpha1>1,
  alpha2>0;
```

$$rhoR := \frac{(-1+R0)\,\sqrt{N}\,\sqrt{\alpha 2}}{\sqrt{\alpha 1}} \qquad (21)$$

We determine also the first two terms in the asymptotic approximations of the elements of SIG for large alpha:

```
> s11b:=map(simplify,convert(asympt(s11,alpha1,1),polynom));
  s12:=s12;
  s13b:=map(simplify,convert(asympt(s13,alpha1,1),polynom));
  s21:=s21;
  s22b:=map(simplify,convert(asympt(s22,alpha1,2),polynom));
  s23b:=map(simplify,convert(asympt(s23,alpha1,2),polynom));
  s31b:=map(simplify,convert(asympt(s31,alpha1,1),polynom));
  s32b:=map(simplify,convert(asympt(s32,alpha1,2),polynom));
  s33b:=map(simplify,convert(asympt(s33,alpha1,1),polynom));
```

$$s11b := \frac{\alpha 1}{\alpha 2\, R0^2} + \frac{\alpha 2\, R0^2 + \alpha 2\, R0 - R0 - \alpha 2 + 1}{\alpha 2\, R0^3}$$

$$s12 := -\frac{1}{R0}$$

$$s13b := -\frac{\alpha 1}{\alpha 2\, R0^2} - \frac{-\alpha 2 + 1 + \alpha 2\, R0 - R0}{\alpha 2\, R0^3}$$

$$s21 := -\frac{1}{R0}$$

$$s22b := \frac{-1 + R0}{R0^2} + \frac{R0^3\, \alpha 2 - \alpha 2\, R0^2 + R0^2 + \alpha 2 - 1}{\alpha 1\, R0^3}$$

$$s23b := \frac{1}{R0^2} + \frac{\alpha 2\, R0^2 - \alpha 2 + 1}{\alpha 1\, R0^3}$$

$$s31b := -\frac{\alpha 1}{\alpha 2\, R0^2} - \frac{-\alpha 2 + 1 + \alpha 2\, R0 - R0}{\alpha 2\, R0^3}$$

$$s32b := \frac{1}{R0^2} + \frac{\alpha 2\, R0^2 - \alpha 2 + 1}{\alpha 1\, R0^3}$$

$$s33b := \frac{\alpha 1}{\alpha 2\, R0^2} + \frac{R0^3\, \alpha 2 - R0 - \alpha 2 + 1}{\alpha 2\, R0^3} \qquad (22)$$